

Modern Programming

John Skaller

December 9, 2015

Contents

I	Felix Overview	4
1	Quick Start	5
1.1	Integers: <code>int</code>	5
1.2	Other integer types	6
1.2.1	Conversions	6
1.2.2	Bitwise operations	9
1.3	Booleans: <code>bool</code>	9
1.3.1	Conditional Expression	9
1.4	Special symbols	9
1.5	Variables: <code>var</code>	10
1.6	Pointers &	11
1.7	Floating Point: <code>double</code>	12
1.7.1	Other floats	12
1.8	Strings: <code>string</code>	13
1.8.1	Escape Codes	13
1.8.2	String Functions	14
1.8.3	Regexps and Regdefs	15
1.9	Tuples	17
1.9.1	Unit tuple	18
1.9.2	Arrays	18
1.10	Unit Sums	18
1.11	Records	19
1.12	Structs	19
1.13	Lists	20
1.14	Pattern Matching and Union	20
1.14.1	Option Type	21
1.14.2	Enumerations <code>enum</code>	22
1.15	Functions	22
1.15.1	Calling functions	23
1.15.2	Overloading	24
1.15.3	Higher Order Functions	24
1.15.4	Polymorphic Functions	25
1.15.5	Conversion Operators	26

1.15.6	Apply operator	27
1.16	Procedures	27
1.17	Generators	28
1.18	Control Flow	28
1.18.1	Conditional	28
1.18.2	Jumps	29
1.18.3	Loops	29
1.18.4	Statement Groups	30
1.19	Yielding Generator	32
1.20	Concepts: type classes	32
1.20.1	Monomorphic Classes	33
1.20.2	Polymorphic Classes	33
2	Interfacing	39
2.1	Embedding C++	39
2.1.1	Lifting types	39
2.1.2	Lifting Values	40
2.1.3	Lifting Functions	40
2.1.4	Fixed Insertions	42
2.1.5	Floating Insertions	43
2.2	Foreign Libraries	44
II	Functional Programming	1
3	Functional Basics	2
3.1	Recursion	2
3.1.1	Reversing a list	2
3.2	More stuff.. . . .	4
3.2.1	Tail Recursion	5
3.2.2	Higher Order Functions	6
3.2.3	Polymorphism	6
3.2.4	Let/in construction	8
3.3	Generalised Folds: Catamorphisms	11
3.4	Maps	11
III	Procedural Programming	15
4	Procedural Basics	16
5	Dynamic Objects	17
6	Reactive Programming	18

<i>CONTENTS</i>	3
IV Active Programming	19
7 Fibres	20
8 Synchronous Channels	21
9 Pipelines	22
10 Coroutines	23
11 Continuations	24
12 Symmetric Lambda Calculus	25
V Concurrency	26
13 Asynchronous Events	27
14 Pre-emption	28
14.1 Pthreads	28
14.1.1 Spawning Pre-emptive Threads	28
14.1.2 Atomics	28
14.1.3 Locks	28
14.1.4 Barriers	28
14.1.5 Concurrently Statement	28
14.2 Pchannels	28
15 Parallelism	29
15.1 Thread Pool	29
15.2 Parallel For Loop	29
VI Categorical Programming	30
16 Category Theory	31
17 Symmetric Categorical Language	32
18 Meta-programming	33
19 Polyadicity: The Holy Grail	34

Part I

Felix Overview

Chapter 1

Quick Start

We must of course begin with the traditional greeting!

```
println$ "Hello World";
```

Here `println` is a procedure which outputs a value to standard output. The argument is of course a literal of type `string`. Calls to procedures must be terminated by a semicolon `;`. You will note, we do not require parentheses around the argument to denote a procedure call! We do not like parentheses much! The `$` sign will be explained in more detail later, but for now you should know it is just a low precedence, right associative application or call operator.

You can run this program from your console or terminal, once Felix is installed, by just typing:

```
flx hello.flx
```

assuming the file `hello.flx` contains the sample code and is in the current directory. Behind the scenes Felix does dependency checking, translates the program to C++, compiles the C++ to a machine binary, and runs it.

It works like Python but it performs like C++.

For more information see <http://felix-lang.org>.

1.1 Integers: `int`

Felix has the usual integer type `int` and literals consisting of decimal digits, and the usual **binary operators** `+` for addition, `-` for subtraction, `*` for multiplication, `/` for division, and `%` for remainder. We also have `<<` to multiply by a power of 2, and `>>` to divide by a power of 2. These have the same semantics

as in C, because, these operators are in fact implemented in the Felix standard library by delegating to C.

```
println$ 42;
println$ (42 + 12) * 90 - (36 / 2 + 1) * 127 % 3;
```

Note you must put spaces around the `-` operator! This is because in Felix `-` is also a hyphen, allowed in identifier names.

We also have the **unary operator** `-` for negation, and for symmetry we also have unary `+` which does nothing.

These operators are just functions so here are some more: the function `str` converts an `int` to a readable string, the function `abs` finds the absolute value of an integer and `sgn` returns `-1` if its argument is negative, `0` if it is zero, and `1` if it is positive.

We also have the usual **comparisons** on integers represented as infix operators: `==` for equality, `!=` for inequality, `<` for less than, `>` for greater than, and `<=` and `>=` for less than or equal and greater than or equal, respectively.

1.2 Other integer types

Felix has a lot of other integer types corresponding to those found in C. Each of these types is distinct, none are aliases! See Table 1.2.

Lexicology

Integers may be specified with a radix, and an underscore may be used between digits as a separator:

```
var x = 0b1111_0011; // binary
var y = 0o777_321;   // octal
var z = 0d998_444;   // decimal
var a = 0xBADE_DAFE; // hex
```

1.2.1 Conversions

All the integer types can be inter-converted with a conversion. This is just the name of the target type, used as a function. For example:

```
var x : long = long 42uz;
var y : int64 = x.int64;
```

The dot notation is just shortcut high precedence reverse application operator much beloved by OO enthusiasts.

Table 1.1: Integer Functions

name	kind	semantics
All Integers		
<code>==</code>	<code>T * T -> bool</code>	equality
<code>!=</code>	<code>T * T -> bool</code>	inequality
<code><</code>	<code>T * T -> bool</code>	less
<code><=</code>	<code>T * T -> bool</code>	less or equal
<code>></code>	<code>T * T -> bool</code>	greater
<code>>=</code>	<code>T * T -> bool</code>	greater or equal
<code>+</code>	<code>T * T -> T</code>	addition
<code>-</code>	<code>T * T -> T</code>	subtraction
<code>*</code>	<code>T * T -> T</code>	multiplication
<code>/</code>	<code>T * T -> T</code>	quotient
<code>%</code>	<code>T * T -> T</code>	remainder
<code><<</code>	<code>T * T -> T</code>	multiplication by power of 2
<code>>></code>	<code>T * T -> T</code>	division by power of 2
<code>-</code>	<code>T -> T</code>	negation
<code>+</code>	<code>T -> T</code>	no op
Signed Integers		
<code>sgn</code>	<code>T -> T</code>	sign
<code>abs</code>	<code>T -> T</code>	absolute value
Unsigned Integers		
<code>\&</code>	<code>T * T -> T</code>	bitwise and
<code>\ </code>	<code>T * T -> T</code>	bitwise or
<code>\^</code>	<code>T * T -> T</code>	bitwise exclusive or
<code>~</code>	<code>T * T -> T</code>	bitwise complement

Table 1.2: Felix Integer Types

Felix	C	Suffix
Standard signed integers		
tiny	char	42t
short	short	42s
int	int	42
long	long	42l
vlong	long long	42ll
Standard unsigned integers		
utiny	unsigned char	42ut
ushort	unsigned short	42us
uint	unsigned int	42u
ulong	unsigned long	42ul
uvlong	unsigned long long	42ull
Exact signed integers		
int8	int8_t	42i8
int16	int16_t	42i16
int32	int32_t	42i32
int64	int64_t	42i64
Exact unsigned integers		
uint8	uint8_t	42u8
uint16	uint16_t	42u16
uint32	uint32_t	42u32
uint64	uint64_t	42u64
Weird ones		
size	size_t	42uz
intptr	uintptr_t	42p
uintptr	uintptr_t	42up
ptrdiff	ptrdiff_t	42d
uptrdiff	ptrdiff_t	42ud
intmax	intmax_t	42j
uintmax	uintmax_t	42uj
Addressing		
address	void*	
byte	unsigned char	

1.2.2 Bitwise operations

For unsigned integers only, bitwise operations are supported. These are `\&` for bitwise and, `\|` for bitwise or, `\^` for bitwise exclusive or, and `~` for bitwise complement.

1.3 Booleans: `bool`

The result of a comparison is a new type, `bool` which has two values, `false` and `true`. You can print booleans:

```
println$ 1 < 2;
```

Felix also has a special statement for asserting that a boolean value is true:

```
assert$ 1 < 2;
```

If the argument of an `assert` is false, then if control flows through it, the program is terminated with an error message.

Booleans support the usual **logical operators**, but in Felix they are spelled out. Conjunction is spelled `and`, whilst disjunction is spelled `or`, implication is spelled `implies`. Of course we also have negation `not`.

1.3.1 Conditional Expression

With `bool` and `int` we can demonstrate the conditional expression:

```
println$
  if 1 < 2 then "less"
  elif 1 > 2 then "greater"
  else "equal"
  endif
;
```

Note that `bool` also forms a **total order** and can be compared, we have `false < true` so that `a==b` means *a* is equivalent to *b*, we can also say that *a* is true if and only if *b* is true. It turns out inequality `a!=b` is the same as exclusive or. Be careful though, since if *a* is less than or equal to *b*, written `a<=b` this actually means that *a* implies *b* logically!

1.4 Special symbols

Felix has a very rich set of symbols. This is because, as well as the usual ascii-art symbols, it also supports almost the complete set of \TeX , \LaTeX , and \AmSTeX

symbols.

TeX symbols start with an backslash and are followed by a sequence of letters. These symbols display in typeset format when using the `flx_web` web server.

For example the following code:

```
var \alpha = 1;
fun \Gamma (x:int)=> x * x;
println$ \Gamma\alpha;
```

should be typeset as

```
var  $\alpha$  = 1;
fun  $\Gamma$  (x:int)=> x * x;
println$  $\Gamma\alpha$ ;
```

1.5 Variables: var

Felix is a procedural programming language, so it has variables! A variable denotes an addressable, mutable, storage location which in Felix, like C, is called an *object*.

```
var x = 1;
var y = 2;
var z = x + 2 * y;
println$ z;
z = 2 * x + y;
println$ z;
```

This code shows variables can be used to factor expressions into a sequence of assignments. We assign variable x the value 1, variable y the value 2, add x to twice y and put the result in variable z , then print it.

Then we assign perform a different calculation and assign that value to z and print it.

What appears to be a variable initialisation is actually equivalent to a definition of an uninitialised variable followed by an assignment.

```
var x: int;
x = 1;
```

The first statment reserves uninitialised store of type `int` named x and the second assigns a value to it. Be very careful with variables, initialised or not! Felix has setwise scoping rules, which are similar to C's function scope used for labels. This means in a scope, you can refer to any symbol defined *anywhere* in that scope. We shall see this is useful for recursive functions because it

eliminates the need for forward declarations. However the following code has undefined behaviour:

```
println$ x;
var x = 1;
```

There is no syntax error, no type error, and no lookup error in this code. The programmer used an uninitialised variable: even though the variable is assigned a value, it is done too late.

Strangely, this code has deterministic behaviour:

```
println$ x;
var x = "Hello";
```

but it may not do what you expect! It prints nothing! The reason is simple enough: when Felix creates a variable it is first initialised with its C++ default constructor. Since a Felix `int` is literally a C++ `int` the default constructor exists, but it is said to be trivial, meaning, it does nothing. This is to improve performance, in the case the first use of the variable will be to assign a value to it: there's no point putting a value in there and then overwriting it!

On the other hand Felix `string` type is just C++ `::std::basic_string<char>` and its default initialiser sets the string to the empty string `""`. That's what the code above prints!

1.6 Pointers &

Since variables are mutable addressable objects we can take the address of a variable:

```
var x = 1;
var px : &int = &x;
var y = *px;
px <- 42;
println$ x,y;
```

Here, the type of a pointer to `int` is written `&int`, the address of a variable `x` is written `&x`, the dereference operation applied to the pointer `px` is written `*px`.

The store operation `px <- 42` is a core operation. In particular assignment is just a shorthand for storing at the address of a variable:

```
x = 42; // means
&x <- 42;
```

Note that in principle the `&` symbol is *NOT* an operator. A variable is in fact a constant, its not variable at all. It's just that the constant is the address of a

storage object. What's stored *at* the address, or *in* a variable may change, the variable itself does not.

1.7 Floating Point: double

Felix also provides a model of C++ type `double` with the usual operators. This is a double precision floating point type which usually follows IEEE standard. You can write a double precision literal in the usual way. Felix follows ISO C-99 for floating point literals.

A set of useful functions is also provided, corresponding to those found in C-99 header file `math.h`.

```
var x = 1.3;
var y = 0.7;
assert sqrt (sqr (sin x) + sqr (cos y)) - 1.0 < 1E-6;
```

Note there is a special caveat with floating point arithmetic. In Felix, `-` has higher precedence than `+`. This means that:

```
var x: double = something;
var y: double; something_else;
assert x + y - y == x;
```

because the subtraction is done first. This can make a difference for integers too, if a calculation overflows, but most floating point operators are not associative: order matters!

Similarly, division has a higher precedence than multiplication!

Floating reals are totally ordered and support exact comparisons. However these operations are not numerically sound, they're based on the underlying finite representation.

Floats also provide checks for `nan` pseudo value, as well as `+inf` and `-inf`.

1.7.1 Other floats

As well as `double` Felix provides single precision `float` and extended precision `ldouble` based on `float` and `long double` respectively, with the equivalent operators.

It also supports complex numbers with Cartesian representation based on C++ complex forms. The types are `fcomplex` based on `float`, `dcomplex` based on `double`, and `lcomplex` based on `ldouble`.

In addition Felix provides double precision based `quaternion` type.

1.8 Strings: string

Felix uses C++ strings for its own strings for compatibility. String literals have 6 forms following Python. Strings not spanning multiples lines can be enclosed in either single or double quotes. Strings spanning multiple lines may be enclosed in tripled single or double quotes.

```
var ss1 = 'Short String';
var ss2 = "Short String";
var ls3 = """
A poem may contain
many lines of prose
""";
var ls4 = '''
Especially if it is written
by T.S. Elliot
''';
```

Note that the triple quoted strings contain everything between the triple quotes, including leading and trailing newlines if present.

Strings can be concatenated by writing them one after the other separated by whitespace.

```
var rose = "Rose";
var ss5 =
    "A " rose ", "
    "by another "
    "name."
;
```

Note that concatenation works for string expressions in general, not just literals.

1.8.1 Escape Codes

Special escapes may be included in strings. The simple escapes are for newline, `\n`, tab `\t`, form feed `\f`, vertical tab `\v`, the escape character `\e`, `\a` alert or bell, `\b` backspace, `\'` single quote, `\"` double quote, `\r` carriage return, `\\` backslash (slosh).

These can be used in any simple string form. Note carefully each is replaced by a single character. This includes `\n`, even on Windows.

In addition Felix provides `\xXX` where each X is one of the hex digits 0123456789, *ABCDEF*, or, *abcdef*. The hex escape is at most two characters after the `x`, if the second character is not a hex digit, the escape is only one character long, the sequence is replaced by the char with ordinal value given by the hex code.

Felix also provides decimal and octal escapes using `\dDDD` and `\o000` respectively, with a 3 character limit on the decoder. Note carefully Felix does NOT provide C's octal escape using a 0 character. Octal is totally archaic.

Felix also provides two unicode escapes. These are `\uXXXX` and `\UXXXXXXXX` which consist of up to 4 and up to 8 hex digits exactly. The corresponding value is translated to UTF-8 and that sequence of characters replaces the escape. The value must be in the range supported by UTF-8.

Felix also provides raw strings, in which escapes are not recognised. This consists of the letter `r` or `R` followed by a single or triple quoted string with double quote delimiter. You cannot use the raw prefix with single quoted strings because the single quote following a letter is allowed in identifiers.

1.8.2 String Functions

Felix has a rich set of string functions. The most important is `char`, which returns a value of type `char`. If the string argument has zero length, the character with ordinal value 0 is returned, otherwise the first character of the string is returned. Again, following Python, Felix does not provide any character literals!

Comparisons

We provide the usual comparison operators.

Length

Felix also provides the most important function `len` which returns the length of a string. The return type is actually `size` which is a special unsigned integer type corresponding to ISO C's `size_t`.

Substring

Felix can fetch a substring of a string using Python like convenions.

```
var x = "Hello World";
var copied = x.[to];    // substring
var hello = x.[to 5];   // copyto
var world= x.[6 to];    // copyfrom
var ello = x.[1 to 5];  // substring
var last3 = x.[-3 to];  // substring
```

The first index is inclusive, the second exclusive. The default first position is 0, the default last position is the length of the string. If the range specified goes

off either end of the string it is clipped back to the string. If the indices are out of order an empty string is returned.

A negative index is translated to by adding the string length.

The substring function is defined so it cannot fail. The name of the actual library function called by this notation is shown in the corresponding comment.

Index

To fetch a single character use:

```
var x = "Hello world";
var y : char = x.[1]; // subscript
```

If the index is out of range, a character with ordinal 0 is returned. Negative indices are translated by adding the string length. The index function cannot fail.

1.8.3 Regexps and Regdefs

Felix provides Google's [Google RE2](#) engine for regular expressions. The basic syntax and capabilities are a subset of Perl's PCRE, only RE2 actually works correctly and performs well. RE2 does not support backreferences.

Basic Matching

A regexp can be compiled with the `RE2` function. Matching is done with the `Match` function. `Match` only supports a complete match. There's no searching or partial matching. Instead, just use repeated wildcards as shown.

```
var r = RE2(" *([A-Za-z_] [A-Za-z0-9]*) .*");
var line = "Hello World";
var maybe_subgroups = Match (r, line);
match maybe_subgroups with
| #None => println$ "No match";
| Some a =>
    println$ "Matched " + a.1;
endmatch;
```

Streamable matching

You may want to match more than one instance of a pattern in a string. For example, you may want to capture each word in a line of text. This can be done by iterating over a regex like the following


```
var r2 = RE2("\w+"); // try to match a word
var sentence = "Hello World";
for x in (r2, sentence) do
    println$ x.0;
done
```

Regular definitions

Regular expressions are quoting hell. Luckily Felix provides a solution: regular definitions.

```
regdef lower = charset "abcdefghijklmnopqrstuvwxyz";
regdef upper = charset "ABCDEFGHIJKLMNOPQRSTUVWXYZ";
regdef digit = charset "0123456789";
regdef alpha = upper | lower;
regdef cid0 = alpha | "_";
regdef cid1 = cid0 | digit;
regdef cid = cid0 cid1 *;
regdef space = " ";
regdef white = space +;
regdef integer = digit+;
```

These are some basic definitions. Note that **regdef** introduces a new syntax corresponding with the notation usually used for regular expressions.

This is called a DSSL or Domain Specific Sub-Language. Its not a DSL, because that's a complete new language, rather the *sub* suggests its an extension of normal Felix. The extension is written entirely in user space. Now to use these definitions:

```

// match an assignment statement
regdef sassign =
  white? "var" white?
  group (cid) white? "=" white?
  (group (cid) | group (integer))
  white? ";" white?
;

var rstr : string = sassign.Regdef::render;
var ra = RE2 rstr;
var result = Match (ra, " var a = b; ");
println$
  match result with
  | #None => "No match?"

  | Some groups =>
    "Assigned " + groups.1 +
    if groups.2 != "" then
      " from variable " + groups.2
    else
      " from integer" + groups.3
    endif
  endmatch
;

```

Note that the regdef kind of variable must be converted to a Perl regexp in a string form using the `render` function.

1.9 Tuples

Felix has an structurally typed product where components are accessed by position, commonly called a *tuple*. Tuples can be constructed using the non-associative n-ary comma operator and accessed by using a plain decimal integer as the projection function:

```

var x = 1, "Hello", 42.7; // type int * string * double
var i = x . 0;
var h = x . 1;
var d = x . 2;

```

Tuple is just another name for Cartesian product. They allow you to pack several values together into a single value in such a way that you can get the components you put in out again.

We shall see tuples are vital for functions, since functions can only take a single argument. To work around this fact, we can pack multiple values together using

a tuple.

1.9.1 Unit tuple

There is a special tuple with nothing in it called the unit tuple, written:

```
var u : unit = ();
var u2 : 1 = ();
```

It is a `unit` type.

1.9.2 Arrays

If all the elements of a tuple are the same type, its called an *array*. In this case the accessor index, or projection, can be an expression of `int` type.

```
var x : int ^4 = 1,2,3,4;
for var i in 0 upto x.len.int - 1 do
  println$ x . i;
done
```

You will also see the use of a low level `for` loop here, and the use of the `len` function to get the array length. The `len` function returns a value of type `size` but the loop variable `i` is an `int`, so we have to convert it.

1.10 Unit Sums

The type of the array in the last section is given by

```
int ^ 4 = int * int * int * int
```

This is the usual meaning of the exponential operator \wedge : raising to some power n means multiplication of n copies.

However you may be surprised to learn that in Felix, `4` is a type, and

```
4 = 1 + 1 + 1 + 1
```

The type `1` here is also called `unit` and is just the type of the empty tuple `()`. Now, you have seen that the cartesian product type is denoted using the n -ary non-associative operator `*`, so it is not so surprising that the type denoted by the n -ary non-associative operator `+` is called a sum. So, the type `4` is called a *unitsum* because it is the sum of units.

Values of a unit sum just represent cases. The notation is ugly:

```
var x : int ^4 = 1,2,3,4;
var third = case 2 of 4;
println$ x.third; // the *true* projection
```

especially as it's zero origin. When you use an integer array index, it has to be bounds checked at run time. However if you use a unit sum index, the check is done by the type system at compile time.

There is a very commonly used unit sum. You already know it: `bool` is nothing more than another name for type 2 and the special words `false` and `true` are just aliases for `case 0 of 2` and `case 1 of 2` respectively!

1.11 Records

A record is a tuple with named fields.

```
var r : (a:int, b:string) = (a=42, b="hitch-hiker");
println$ r.a, r.b;
```

Records support advanced features including row polymorphism with scoped labels.

1.12 Structs

A struct is a nominally typed product with named fields.

```
struct X {
  a:int;
  b:string;
  fun show() => self.b + " " + self.a.str;
};
var x = X(42, "World");
x.show;
```

Functions and procedures can be included in a record but are actually defined outside it with a curried argument named `self`. It has the type of the record for a function, and a pointer to the record for a procedure. So the above is equivalent to:

```
struct X {
  a:int;
  b:string;
};
fun show(self:X) => self.b + " " + self.a.str;
```

1.13 Lists

A list is a variable length sequence of values of the same type. An empty list of `int` is denoted `Empty[int]`. Given a list you can create a new one with a new element on the front using the constructor `Cons` as follows:

```
var x = Empty[int];
x = Cons (1,x);
x = Cons (2,x);
x = Cons (3,x);
var y = Cons (3, Cons (2, Cons (1, Empty[int])));
```

Of course this is messy! Here is a better way:

```
var z = list (3,2,1);
```

This method converts a tuple to a list. You can add two lists together, prepend a value, or add a value to the end of a list with the infix `+` operator:

```
var x = list (1,2,3);
x = 1 + x + x + 42;
```

Take care that `+` associates to the left and you don't accidentally add two integers together! There is a second operator you can use as well which is right associative and prepends an element to a list:

```
var x = 3 ! 2 ! 1 ! Empty[int];
x = 42 ! x;
```

You can use the `len` function to find the length of a list, and test if an element is in a list using the `in` operator:

```
var x = list (1,2,3);
assert len x in x; // 3 is in the list!
```

Lists in Felix are purely functional data structures: you cannot modify a list. All the nodes in a list are immutable, which means when you prepend an element *A* to a list *L*, and then prepends an element *P* to the same list *L*, the tail of the list is shared. Lists can be passed around efficiently without copying.

When some prefix of a list is no longer accessible because the function prepending the prefix returns without saving the list, the prefix elements will be removed automatically by the Felix garbage collector.

1.14 Pattern Matching and Union

A list is actually defined like this:

```

union list[T] =
  | Empty
  | Cons of T * list[T]
;

```

The `union` construction defines a nominally typed sum. The words `Empty` and `Cons` are called type constructors. This union is also polymorphic! This means you can make lists of any type. The first case is just an empty list. The second case, `Cons` joins together an element of type `T` and another list. In other words it creates a new list by adding a new element onto the front. A type like this is called an [inductive type](#).

A list can be taken apart with a pattern match:

```

var x = list (3,2,1);
println$
  match x with
  | #Empty => "Empty"
  | Cons(v, tail) => "first element " + v.str
  endmatch
;

```

1.14.1 Option Type

Another commonly used union type is the option type `opt`:

```

union opt[T] =
  | None
  | Some of T
;

fun maybe_divide (num:int, denom:int) =>
  if denom == 0 then None[int]
  else Some (num/denom)
;

println$
  match maybe_divide (10,0) with
  | #None => "Divide by Zero"
  | Some j => "Quotient " + j.str
  endmatch
;

```

1.14.2 Enumerations `enum`

If all the constructors of a union have no arguments an alternate form can be used:

```
enum colour = red, green, blue; // equivalent to
union colour = red | green | blue;
```

1.15 Functions

We have enough preliminaries now to finally introduce functions. Without further ado, here are some basic functions:

```
fun twice (x:int) : int => x + x;
fun thrice (x:int) :int => twice x + x;
```

Functions also have a more expanded form:

```
fun trickdiv (num:int, denom:int) :int =
{
  var y = if denom == 0 then 1 else denom endif;
  return xnum / y;
}
```

There is a rule for functions:

functions defined with the `fun` binder may not have observable side effects

The rule is relaxed; that is, not enforced, to permit debugging, profiling, coverage checking, etc.

Pattern Syntax for Functions

Functions can also be defined with an abbreviated form which merges the definition with a pattern match:

```
fun len[T] : list[T] -> int =
  | #Empty => 0
  | Cons (_, tail) => 1 + len tail
;
```

This is just shorthand for

```

fun len[T] (x:list[T]): int =>
  match x with
  | #Empty => 0
  | Cons (_, tail) => 1 + len tail
  endmatch
;

```

1.15.1 Calling functions

In Felix you can call a function using operator whitespace:

```

fun twice (x:int) : int => x + x;
println$ twice 42;

```

This is the usual prefix mathematical notation. In Felix, parentheses are not required around arguments to make a function call, they are just used for grouping. Some programmers also like postfix notation and you can use that too:

```

fun twice (x:int) : int => x + x;
fun thrice (x:int) :int => twice x + x;
println$ thrice 42.twice;

```

Here, `twice` is called first, because operator `.` has a higher precedence than operator whitespace. Both operators are left associative! Operator dot is called *reverse application* because the argument is written first, then the function to apply.

Felix also has operator dollar `$` which is a very low precedence right associative operator:

```

println$ k $ h $ g 42.f;
println ( k ( h ( g ( f 42))));

```

and there is also a left associative low precedence reverse order operator too:

```

h $ g 42.f |> k |> println;
println ( k ( h ( g ( f 42))));

```

This is called *reverse pipe application* and has an even lower precedence than forward dollar application.

There is one more application operator!

```

fun hhgttg() => 42;
h $ g #hhgttg.f |> k |> println;

```

Operator hash `#` just applies a function to the unit tuple `()`, it is often used for constant functions. It is a very high precedence operator.

It's no wonder they wanted to spacedoze the Earth to build a freeway.

1.15.2 Overloading

You can define two functions with the same name, this is never an error. However to apply a function with the same name as another, it must be possible to distinguish between them.

Overloading provides one method for doing this:

```
fun twice (x:int) => x + x;
fun twice (x:double) => x + x;
println$ twice 42, twice 42.1;
```

In the above example, the first application in the `println` argument tuple calls the first function, because the argument is of type `int`, and the second application calls the second function because the argument is of type `double`.

Selecting functions based on matching the argument and parameter types like this is called overload resolution. Note that unlike C++ there are no automatic conversions in Felix. The argument and parameter type must match exactly.

Overloading first looks in the scope of the application. If matching fails to find any candidate functions, it proceeds to the next outer level, until there are no more levels left.

1.15.3 Higher Order Functions

In Felix, a function can become a first class value. This value is called a closure. Functions that accept functions as arguments, or return functions, are called higher order functions or *HOF*'s.

```
fun twice(x:int) => x + x;
fun both (x:int, y:int, g: int -> int) => g x, gy;
println$ both (3,7,twice);
```

The function `both` applies its argument `g` to both `x` and `y`, returning the pair of results. On the other hand here:

```
fun increment (x:int) : int * int -> int =
{
  fun add (y:int) => x + y;
  return add;
}
println$ increment 3 42; // 45
var add3 = increment 3;
println$ add3 42; // 45
```

the `increment` function returns another function which increments its argument `y`, when applied, by the argument `x` first passed to it. The variable `add3` binds the

first argument `x`, and is another function which will add that `x` to its argument when applied.

The function `increment` is said to have *arity 2* because it appears to accept two arguments. In fact, there is no such thing as a function accepting two arguments.

The function of higher arity is sometimes said to be *curried* named for theorist Howard Curry, although some prefer Indian food. If a function returns a function which is not immediately applied to an argument it is said, surprisingly, to be *partially applied*. When Felix does overload resolution it can take into account as many arguments as are provided for a curried function application. It finds candidates based on the first argument first, then tries to whittle down the list based on the next argument, and so on.

1.15.4 Polymorphic Functions

In Felix, you can write functions that work with a family of types.

```
fun diag[T] (x:T) => x, x;
```

This is the famous diagonal function, which makes two copies of its argument. It works for any type, so it is a polymorphic function. In general such functions can *only* do structural manipulations so they are of limited utility.

Unless, that is, they're HOFs! For example:

```
fun add3[T] (x:T, y:T, z:T, add:T * T -> T) =>
    add (x, add (y,z))
;
```

Overloading Polymorphic Functions

Polymorphic functions can be overloaded too. However unlike ordinary functions, more than one function might match. For example:

```
fun f[U,V] (x: U, y: V) => y,x;
fun f[W] (x:W, y:int) => y+1,x;
println$ f (2,4); // which f?
```

A function matches if the parameter type can be specialised to the argument type by substituting some type expression for each of the function's type variables. In the above case

$$\begin{aligned} U &\rightarrow \text{int} \\ V &\rightarrow \text{int} \end{aligned}$$

causes the first function to match and

$$W \rightarrow \text{int}$$

causes the second one to match.

In this case the most specialised function is chosen if there is one. Here, the first function is clearly more general because the substitution:

$$\begin{aligned} U &\rightarrow W \\ V &\rightarrow \text{int} \end{aligned}$$

into the type of the first function's parameter yields the second function's parameter type. Since there is no substitution in the other direction, the second function is strictly more specialised and so it is selected.

The process of finding a substitution which makes types with type variables equal is called [unification](#). The substitution effecting equality is called a *unifier*, and the two terms rendered equal are said to be *unified*. There can be more than one unifier, a most general unifier is one which can produce all the other unifiers by some substitutions. If there is a most general unifier it is unique up to changing variable names.

1.15.5 Conversion Operators

Felix has a special shortcut for defining conversions:

```
typedef cart_complex = (x:double, y:double);
typedef polar_complex = (r:double, theta:double);

ctor cart_complex (z: polar_complex) =>
  (x=z.r * cos z.theta, z.r * sin z.theta)
;

ctor polar_complex (z: cart_complex) =>
  (r = sqrt (z.x.sqr + z.y.sqr), theta = tan (y / z))
;

ctor cart_complex (x:double) => (x=x,y=0.0);
ctor polar_complex (x:double) => (r=x, theta=0.0);

var z1 = cart_complex 0.5;
var z2 = polar_complex z1;
println$ z2;
```

The binder `ctor` must be followed by the name of a type. A `typedef` alias will do. The definition is expanded:

```
ctor T (x:arg) => expr;

//equivalent to:
fun _ctor_T (x:arg) : T => expr;
```

When Felix finds an application `fa` where `f` is a simple name, it first tries to find a function named `f`. If a type name is found instead of a function, Felix looks again for a function named `_ctor_f` where `f` is the type name.

1.15.6 Apply operator

When Felix finds an application `fa` where `f` is a simple name, it first tries to find a function named `f`. If a value of some type `F` is found instead, Felix looks again for a function named `apply` with a tuple argument of type `F * A` where `A` is the type of the argument. For example:

```
fun apply (x:string, y:string) => x + y;
println$ "Hello" " " "World";

fun apply (x:double, y:double) => x * y;
fun apply (x:int, y:double) => x * y;
var x = 2.3;
println$ 2 x + 4 x x;
```

This allows objects of any type other than functions to also act as functions in a user defined way.

1.16 Procedures

A *procedure* is like a function which returns no value and is allowed to have side effects.

```
proc printint(x:int)
{
  println$ "An integer " + x.str;
}
```

This is in fact syntactic sugar for

```
fun printint(x:int) : void =  
{  
  println$ "An integer " + x.str;  
}  
  
printit 42;
```

The pseudo type `void` is the type you use when there is no value.

Procedures with unit argument type have a special call notation, the unit argument `()` can be dropped:

```
proc print42()  
{  
  println$ "The world is coming to an end";  
}  
  
print42; // no () required
```

You're right, I hate excess parens!

1.17 Generators

There is one further special kind of function, called a *generator*. A generator returns a value and may have a side effect.

```
var counter = 0;  
gen fresh() : int =  
{  
  ++counter;  
  return counter;  
}  
  
println$ #fresh, #fresh; // 1,2
```

The term generator comes from the exemplar generator, namely the random number generators.

1.18 Control Flow

Felix has a rich control flow architecture. Here are some control flow constructs.

1.18.1 Conditional

Here is the expanded procedural conditional branch.

```
begin
  if cond do
    something;
  elif cond2 do
    something_more;
  else
    last; resort;
  done
end
```

There are also several short forms.

```
if cond goto lab;
if cond return;
```

1.18.2 Jumps

The usual goto and target label.

```
begin
  var i = 1;
  start:> // label
  call println i; // procedure call
  ++i; // increment i
  if i < 10 goto start; // conditional jump
end
```

1.18.3 Loops

Low level inclusive loops are flat code. Jumping in and out with goto is permitted. The control variable can optionally be declared in a loop, and exists in the whole containing scope.

```
begin
  for var i:int in 0 upto 10 do // inclusive
    println$ i; // procedure call
  done
  for i in 10 downto 0 do // inclusive
    println$ i; // procedure call
  done
```

Here is the usual while loop and its negation the until loop.

```

begin
  var i = 0;
  while i < 10 do
    println$ i;
    ++i;
  done
  until i == 0 do
    println i;
    --i; // decrement
  done
end

```

And here are loops using iterators, note that the control variable is auto declared:

```

// inclusive subrange of int iterator
for j in 0..10 do
  println$ j;
done

// exclusive subrange of int iterator
for j in 0..<10 do
  println$ j;
done

// array iterator
for j in (1,2,3,4,5,6,7,8,9,10) do
  println$ j; // array iterator
done

// list iterator
for j in list (1,2,3,4,5,6,7,8,9,19) do
  println$ i; // list iterator
done
for (i=0; i<10; ++i;) do // C style
  println$ i;
done
end

```

Iterator based loops are implemented using yielding generators as explained in the next section.

1.18.4 Statement Groups

Statements can be grouped using the `do .. done` construction. The body of a loop is a single statement, the `do` group is used to present the parser many

statements considered as one. Jumps into and out of a `do` group are allowed, they do not represent a scope.

There is another group called a *block* which does represent a scope: the `begin .. end` construction. You can jump out of a block, but you cannot jump in, and you cannot return.

There is a reason for this:

```
var doit = { println$ "Hello"; println$ " World"; };
call doit ();
```

The expression in curly braces `{}` is an anonymous procedure of type $1 \rightarrow 0$. To call this procedure, we must apply it to a value of type unit, and there is only one such value, namely `()`.

The `begin..end` construction is just an anonymous procedure which is then applied to unit immediately, so that

```
{ println$ "Hello"; println$ " World"; } ();
```

is equivalent to

```
{ println$ "Hello"; println$ " World"; };
```

which is equivalent to

```
begin println$ "Hello"; println$ " World"; end
```

only we don't need the trailing semi-colon `;`.

So because this is a procedure, a `return` statement would just return from the anonymous procedure, and not the containing procedure. To work around this you can name the procedure from which you wish to return using a `return from` statement:

```
proc outer (x:int)
{
  {
    println$ "Inner";
    if x > 0 do return from outer; done
  };
  println$ "Negative";
}
```

Do not try to return from a procedure which is not active, all hell will break loose.

1.19 Yielding Generator

Here is a way to write and use what is called an *yielding generator*. This is a generator utilising a `yield` statement.

```
begin
  gen down (var start:int) () =
  {
    for i in start downto 0 do
      yield i;
    done
    return i;
  }

  var it = down 10;
  var x = #it;
  while x >= 0 do
    println x;
    x = #it;
  done
end
```

A yielding generator must be a function of type $1 \rightarrow T$ for some type T . In this case `down 10` has that type. A closure over the generator must be assigned to a variable to hold the state, named `it` in the example.

The closure stored in the variable is then called by applying it to the unit value `()`, written like `#it` here because I hate parens. This causes the generator to run until it yields a value or returns.

The next call to the generator causes it to proceed from where it left off after a yield, or to do the terminating return again if necessary. So yielding generators can always be called infinitely producing a stream of values.

The caller and the generator are coroutines which swap control back and forth as required separately by each, that is, they are both masters. The generator is a so-called *push master* and the client is a *pull master*.

We will see shortly that Felix has different techniques to implement fully general coroutines.

1.20 Concepts: type classes

Felix provides a way of systematically organising functions and operators known as *type classes*, modelled on the construction of the same name in Haskell.

1.20.1 Monomorphic Classes

A nonpolymorphic `class` is just a closed namespace:

```
class X
{
  fun f(x:int) => x + 1;
  fun g(x:int) => x * x;
}

println$ X::f 42, X::g 42
```

As well as using explicit qualification, you can open a class:

```
open X;
fun g(x:int) => "Hello " + x.str;
println$ f 42, g 42; // X::f, root::g
```

When a class is opened it occupies a shadow lookup space just behind the space into which it is opened. Therefore the local `g` above hides `X::g`. The global namespace is called `root` in case you need explicit qualification.

When you open a class inside another, the symbols of the former are available for definition of the latter, but they are not available to clients of the latter. Thus opening is not transitive.

```
class A { fun f(x:int) => x; }
class B { open A; fun g (x:int) => f x; }
type-error println$ B::f 42; // no f in B
```

If you want to include the symbols of one class in another, use `inherit` instead:

```
class A { fun f(x:int) => x; }
class B { inherit A; fun g (x:int) => f x; }
println$ B::f 42; // OK, f is in B
```

1.20.2 Polymorphic Classes

When a class is polymorphic things get more interesting. Polymorphic classes are basically collections of template declarations and definitions.

```

class AddGroup[T]
{
  virtual fun == : T * T -> bool;
  fun != (x:T, y:T) => not (x == t);

  virtual fun + : T * T -> T;
  virtual fun zero : 1 -> T;
  virtual fun neg : T -> T;
  fun - (x:T, Y:T) : T => x + neg y;

  axiom negation (x:T): x + neg x == zero();

  private fun mulby (x:T, count: int) : T =
  {
    var res = zero();
    for i in 0 ..< count do res = res + x; done
    return res;
  }
  fun * (x:T, count: int) : T => mulby (x,count);
}

```

The above class specifies an additive group. The specification is entirely abstract because it is given entirely in terms of functions. The four **virtual** functions are undefined here and form the *basis* of the abstraction.

The inequality, subtraction, and multiplication functions shown are defined in terms of the abstract basis. The **mulby** function is marked as **private** and is only a local helper function, it will not be exported as part of the class interface.

The **axiom** is a semantic constraint on the behaviour of the functions.

As you see the class above has "holes" in it, namely the undefined virtual functions. Here is how to provide definitions:

```

struct MyInt { v:int; } // to avoid conflict with std lib

instance AddGroup[MyInt]
{
  fun == (x:MyInt, y:MyInt) => x.v == y.v;
  fun zero () => MyInt (0);
  fun + (x:MyInt, y:MyInt) => MyInt (x.v + y.v);
  fun - (x:MyInt) => MyInt (-x.v);
}

```

To use this, it is convenient to open our class for our type:

```
open AddGroup[MyInt];  
  
var x = MyInt(1);  
var y = MyInt(42);  
var z = x + -y * 3 - MyInt(74) != AddGroup[MyInt]::zero();
```

Note that we used explicit qualification for the `zero` function although it is not necessary in this case. In general, functions with the same name taking a unit argument have to be explicitly qualified to distinguish them, there is no type distinction in the argument for overload resolution to use.

Two Phase Binding

Felix uses a lookup model which provides two phases of binding. In the first phase, all binding is polymorphic. Bindings to virtual functions may occur in this phase.

In the second phase, the whole program is monomorphised, eliminating all type variables. This phase is sometimes called *instantiation*. When an application or call of a virtual function is seen, Felix searches for an instance matching the monomorphic type parameters and replaces the virtual call with a concrete one, or bugs out if no instance is found.

Polymorphic Recursion

You should note that instance functions may themselves contain virtual calls, and this permits the instantiator to implement polymorphic recursion. That is, a limited form of *second order polymorphism* is available in Felix, but only via type classes. Polymorphic recursions must either bottom out on monomorphic types, or reduce to monomorphic recursion after a finite number of steps. The current compiler does not check this in advance, and instead just uses an expansion limit.

Table 1.3: Summary of Types

Primitive		
integers		
floats		
strings		
regexps		
Nominal		
unions		
structs		
cstructs		
Structural Products		
unit	1	
tuples	$T_0 * T_2 * \dots * T_{n-1}$	
records	$(f_0 : T_0, f_1 : T_1, \dots, f_{n-1} : T_{n-1})$	
Structural Sums		
void	0	
bool	2	
unitsums	99	
sums	$T_0 + T_2 + \dots + T_{n-1}$	
variants		
Structural Exponentials		
arrays	T^N	
functions	$D \rightarrow C$	
cfunctions	$D \dashrightarrow C$	
pointers	$\&T$	
Library		
option	<code>opt[T]=Some of T None</code>	option type
list	<code>list[T]=Empty Cons of (T*list[T])</code>	functional list
varray	<code>varray[T]</code>	bounded stable variable length array
darray	<code>darray[T]</code>	unbounded reallocable variable length array
sarray	sparse array	
strdict	string dictionary	
judy array	Judy1, JudyL arrays	
schannel	synchronous channels	
pchannel	asynchronous channels	

Table 1.4: Summary of Binders

Binders	
var	declare and initialise a variable
val	name a value
def	compound assignment, variable binder, and value binder
fun	define a function
cfun	define a function with C function type
gen	define a generator
ctor	define a conversion function
proc	define a procedure or fibre
cproc	define a procedure with C function type
struct	define a struct
cstruct	model an existing C struct
union	define a discriminated union
enum	define an enumeration
class	define a type class
interface	define an object interface
object	define an object constructor
typedef	define an alias for a type
syntax	define new syntax
type..new	define an abstract type
Directives	
include	include a library file
inherit	inherit symbols into a class
open	open a class
requires	specify requirements of C bindings
open syntax	augment parser with new grammar
comment	write a comment
todo	specify future code
private	do not export symbol from class
rename	rename or import a symbol
macro val	define a macro
forall	generate a table

Table 1.5: Summary of Statements

<hr/> Control <hr/>	
goto	unconditional control transfer
goto-indirect	goto address stored in LABEL variable
return	return from procedure or function
return from	return from specified procedure
call	call a procedure
jump	jump to procedure with arguments
yield	suspend generator emitting value
halt	stop program
if..do..elif..else..done	standard conditional
match..with	pattern matching
with	execute statements with temporary binders
trace	trace execution on UDP port
type-error	document compilation error
_svc	Felix scheduler service call
spawn_fthread	schedule fibre
spawn_pthread	schedule pre-emptive thread
read	read channel
write	write channel
branch-and-link	low level control exchange
<hr/> loops <hr/>	
while	execute whilst condition true
until	execute until condition true
for	execute repeatedly with bounded control variable
break	exit specified loop
continue	start next iteration of specified loop
redo	restart current iterator of specified loop
<hr/> Groups <hr/>	
do..done	statement group
begin..end	scoped statement group
perform	single statement
<hr/> Assignment <hr/>	
p<-a;	
v=a;	
v1<->v2;	
v+=a;	
v-=a;	
v*=a;	
v/=a;	
v%=a;	
v&=a;	
v =a;	
v^=a;	
v<<=a;	
v>>=a;	
++v;	
--v;	
<hr/> Plugins <hr/>	
static-link-symbol	
static-link-plugin	
export	
export python	

Chapter 2

Interfacing

2.1 Embedding C++

Felix was specifically designed to allow almost seamless lifting of C++ code. Indeed most of the standard library types are simply types lifted from C++. The provides ABI compatibility with C++ and hence C, although syntactic compatibility is sacrificed in favour of a superior type system.

2.1.1 Lifting types

The `type` statement is used to lift types from C++.

```
type myint = "int";
var x:myint;

type metres = "double";
var m: metres;

type mystring="::std::basic_string<char>"
  requires header '#include <string>'
;
var s: mystring;
```

Template classes can be lifted to polymorphic types too, the notation `?1`, `?2`, etc is used to associate the type variables.

```
type vector[T] = "vector<?1>"
  requires header '#include <vector>'
;
var v : vector[int]; // vector<int>
```


Note that if the type is defined in a C or C++ header you must ensure the header is included in the code Felix generates as shown. The **requires** clause will be explained shortly.

If you need to lift a lot of non polymorphic types at once there is a special shortcut using **ctypes** binder:

```
ctypes int,long,short,double;
```

2.1.2 Lifting Values

Of course, types are useless without values, so Felix provides a way to lift expressions with the **const** binder:

```
const one : int = "1";
const pi : double = "22.0/7";
var z = pi; // rough approx
```

The Felix type of the lifted value must be given. Note the value can be an arbitrary expression.

Note: it is also possible to lift literals by extending the Felix grammar. The Felix grammar is actually defined in the standard library in user space, so extending it can be done in Felix. However it is beyond the scope of this quick introduction to explain the somewhat arcane details.

2.1.3 Lifting Functions

Lifting types is all very well, but they're not very useful unless they can be manipulated. C++ functions can be lifted to functions and procedures like this:

```
fun addup3 : int * int * int -> int = "$1+$2+$3";
fun sin : double -> double = "sin($1)"
  requires header '#include <math.h>'
;
proc coutit[T] : T = "cout << $1;"
  requires header "#include <iostream>"
;
```

Here is a more sophisticated example:

```

type vector[T] = "::std::vector<?1>"
  requires header "#include <vector>"
;
proc push_back[T] : vector[T] * T = "$1.push_back($2)";
fun len[T]: vector[T] -> int = "$1.size()";

type viter[T] = "::std::vector<?1>::const_iterator"
  requires header "#include <vector>"
;
fun deref[T] : viter[T] -> T = "$*1";
fun next[T] : viter[T] -> viter[T] = "$1+1";
fun !=[T] : viter[T] * viter[T] -> bool = "$1!=$2";

fun begin[T] : vector[T] -> viter[T] = "$1.begin()";
fun end[T] : vector[T] -> viter[T] = "$1.end()";

var v : vector[int]; // C++ default init to empty
begin
  var l = list (1,2,3,4); // Felix list
  for elt in l do push_back(v, elt); done
end

println$ "Length = " + v.len.str;

// print STL vector
var it = v.begin;
while it != v.end do
  println$ deref it;
  it = next it;
done

```

There are some things to note here!

First, you may note the functions `begin` and `end` used here. This is bad practice, the names `stl_begin` and `stl_end` would be better. But the question is: why does it work at all? Aren't `begin` and `end` keywords?

The answer is simple: no, they can't be keywords because Felix has no keywords! You may have thought that there were far too many keywords in Felix but now you know the truth. There are none. Zero. Zilch. Felix uses a sophisticated extended generalised scannerless LR parser and the grammar only recognises certain identifiers as special in contexts where they might occur.

The second important thing to note here is that we have cheated and some care is needed to avoid a serious issue. The Felix `vector` we have defined is polymorphic and can be used with any type where the generated C++ type would work, but it must not be used with a Felix type which uses the garbage collector! Unions and nonunit sum types and Felix pointers in particular usually

require garbage collection, and that includes Felix `list` type.

If you put such a type into a C++ vector, the Felix garbage collector will not know it is there, and objects pointed at may appear to be unreachable and be deleted when they're stored in the vector. It is safe to put these types into a vector or other C++ data structure if the pointers can be regarded as weak; that is, if they're also stored somewhere else which ensures they remain reachable whilst the vector is holding them.

The third thing to note is that we have broken a very important protocol in this example: we have modified a value, namely the vector `v`. Although C++ allows modifying values, Felix does not. Felix only allows modification of objects. In particular although `v` in the example is an addressable store and therefore an object, all modifications to objects must be done via the address of the object, that is, by using a pointer. Felix has no concept of lvalue like C++, and no reference types: we use pointers instead.

The problem is that we have used the abstraction provided by the Felix lifting constructs to hide the underlying data type. The compiler cannot know that the type you lifted was not in fact a pointer to an object and the mutators lifted operating through the pointer to modify the object it points to. That would be allowed!

Now let us fix it:

```
proc push_back[T] : &vector[T] * T = "$1.push_back($2)";
fun begin[T] : &vector[T] -> viter[T] = "$1.begin()";
fun end[T] : &vector[T] -> viter[T] = "$1.end()";
```

There! We have to adjust the corresponding usage of course. Note carefully the iterator itself is indeed an abstract pointer. So operations on the vector via the iterators obtained do NOT require any adjustment! Although none are present in this example.

2.1.4 Fixed Insertions

You can put C++ code directly into Felix. Statements can be added with `cstmt` statement. An argument may optionally be added. This is basically an anonymous Felix procedure written in C++.

Arbitrary expressions can be inserted with `cexpr . . endcexpr`, also optionally taking an argument.

The short form `cvar` lifts a variable.

```

cstmt """
#include <iostream>
static int x = 1;
""";

type INT = "int";
fun + :INT * INT -> INT = "$1 + $2";

fun two (): INT =>
  cvar [INT] x + cexpr[INT] 'x' endcexpr
;
cstmt "::std::cout <<$1<<endl;" (two());

```

2.1.5 Floating Insertions

Floating insertions allow you to conditionally include C++ code based on usage. For example:

```

type vector[T] = "::std::vector[?1]"
  requires header "#include <vector>"
;

proc push_back[T] (&vector[T] :pv, v:T) =>
  "$1->push_back($2);"
;

var x : vector[int];
push_back (&x,1);
push_back (&x,2);

```

The **requires** clause creates a dependency, the **header** specifier says it depends on a floating insertion, which if required will end up near the top of the generated C++ header file for your program. This ensures C++ `::std::vector` is in scope.

The requirement is conditional on the type being used, as it is in the `push_back` procedure. On the other hand, since it is used in that procedure, it isn't necessary to also place a requirement on the use of the procedure. Calling `push_back` requires the `push_back` procedure which in turn requires the `vector` type, which in turn requires the floating insertion.

Floating insertions can be named, and can also have their own requirements. All the standard C, C++, and posix headers are named in `std/c/c_headers` and `std/c/cxx_headers`.

```

header iostream = "#include <iostream>";
type vector[T] = "::std::vector[?1]" requires iostream;

```

2.2 Foreign Libraries

Here is a quick guide to embedding a foreign C library `mylib` with functions `mycal1` and `mycal2` on `double`. The first thing of course is to make a binding file:

```
// file mylib.flx
class MyLib
{
  requires package "mylib";
  fun mycal1: double -> double = "mycal1($1)";
  fun mycal2: double -> double = "mycal2($1)";
}
```

Requirements specified in a `class` are inherited by all C bindings in the class. Here is a new requirement on a `package`, and there is no requirement on a header file!

Now you must make another file, the package descriptor, this one is for Linux:

```
Name: mylib
includes: "mylib.h"
cflags: -Imydir
provides_dlib: -lmylib
provides_slib: -lmylib
```

The file must be named `mylib.fpc`, and placed in a configuration database directory the `flx` processor searches. The specified inclusion (`#include "mylib.h"`) will be put in the generated C++ code automatically. Furthermore the include file search flag `-Imydir` will be added to the C++ compiler compilation command line, and the linker flag `-lmylib` added to the compiler linkage command line. In this case the `provides_dlib` implies the library file `libmylib.so` and `provides_slib` implies `libmylib.a`.

This mechanism works on OSX and Windows, and with `gcc`, `clang`, `cl`, and other compilers. The objective is to ensure the Felix code is platform independent, and shove all the platform dependent stuff into a single place. Here's a windows package:

```
Name: mylib
includes: "mylib.h"
cflags: /Imydir
provides_dlib: /DEFAULTLIB:mylib
provides_slib: /DEFAULTLIB:mylib
```

Felix uses a tool called `flx_pkgconfig` to query and manage packages. The tool itself is platform independent and the databases are also language neutral.

Part II

Functional Programming

Chapter 3

Functional Basics

We shall begin our more serious exploration of Felix with functional programming techniques.

3.1 Recursion

We shall begin with some basic routines on lists.

3.1.1 Reversing a list

To reverse a list, we run down the list constructing a new one. At the end, we return the result. We will use several routines for this.

Our first routine takes two arguments, a list `inp` and a list `out` onto which we must prepend the reversed copy of `inp`.

```
fun prepend_reversed[T] (inp:list[T]) (out: list[T]) =>
  match inp with
  | #Empty => out
  | Cons (head, tail) =>
    prepend_reversed tail (Cons (head, out))
  endmatch
;
```

This routine is statically recursive, meaning, there is a control path which causes the routine to be re-invoked. It is also self-recursive, meaning it invokes itself directly.

A *tail call* is the final call in a function. Tail calls can be optimised from a call followed by a return to just a jump. This avoids pushing the return address, then

returning to that address from the called procedure, then immediately returning to the callers return address. Instead, the called procedure just returns to the callers return address directly.

In particular, if the tail call is call to self, known as *self-tail-recursion* then by assigning the call argument to the parameter, we then just jump to the start of the function. In other words, we can optimise tail recursion to a simple loop. This saves more than the space of the return address on the stack, it saves the space required by all the local variables as well. We can reuse these variables, since the old values cannot be accessed.

Our routine `prepend_reversed` is clearly self tail recursive and can be optimised to a simple loop. Tail recursion is also known as *primitive recursion*.

We can use this routine to reverse a list.

```
fun rev[T] (a: list[T]) => prepend_reversed a Empty[T];
```

Now suppose we wish to prepend a list *a* onto a list *b*. How would we do that? Well it's easy:

```
fun prepend_v1[T] (a: list[T]) (b: list[T]) =>
  prepend_reversed (rev a) b
;
```

We just reverse the first list, then prepend it reversed onto the second one. Which suggests there is another way:

```
fun prepend_v2[T] (a: list[T]) (b: list[T]) =>
  rev (prepend_reversed a (rev b))
;
```

Our first routine requires $2a$ heap allocations. The second routine requires $2(a+b)$ heap allocations. Both numbers look bad! This is because we have paid for tail recursion with excess heap allocations. Here is how we can do it with a non-tail recursive routine with only a heap allocations and a stack allocations.

```
fun prepend_v3[T] (a: list[T]) (b: list[T]) =>
  match a with
  | #Empty => b
  | Cons (head, tail) =>
    Cons (head, prepend_v3 tail b)
  endmatch
;
```

This routine is no use if the machine stack is limited and the list *a* is big. It also requires more maximum store than either of the other routines! The whole of the original list *a* is saved on the stack at the time *a* is empty, and as we return we prepend elements to be in reverse order. On the way down the list, we do save the elements to prepend on the way back up. We therefore have a copy of *a* as well as $n = \text{lena}$ machine addresses on the stack.

On the other hand, whilst our `v1` routine reverses the original list, and then pushes it reversed onto the output list, it uses at most *a* heap nodes. As we push the top of the reversed list onto the output, that head element becomes unreachable and can be freed. In fact, the heap allocation and free can be avoided altogether by simply adjusting the pointer to the tail of the top node to point instead at the output list! It is perfectly safe to do this because that reversed list is owned by us and no one else, and it is entirely unreachable at the end of the routine.

Even more surprising, our `v2` routine can be optimised by a trick as well. Since the argument to the tail call is owned by us, we can just modify the list, instead of reallocating it. We can either reverse the actual elements, or, we can swap the next pointers around!

In general, a temporary created by a function can safely mutated. A basic functional programming language does not have the type system support to enable the compiler to prove when it is safe, however such type systems do exist: the typing regime is known as *linear typing* because it is used to track when a value is constructed it is used exactly once. See [Wikipedia](#) for more information.

3.2 More stuff..

```
fun inner_strlist (x: list[int]) =>
  match x with
  // Empty list
  | #Empty => ""

  // One element list
  | Cons (head, #Empty) =>
    str head

  // Two element list
  | Cons (head, Cons (second, #Empty)) =>
    str head ", " + str second

  // More than two elements
  | Cons (head, tail) =>
    str head ", " + inner_strlist tail
  endmatch
;

fun strlist (x: list[int]) => "list (" + inner_strlist x + ")";

println$ strlist (list (1,2,3));
```

Here, the function `inner_strlist` is recursive. It matches the list so it is empty it produces an empty string. If there is only one element, it produces the string representation of that int.

If the list contains two elements, it produces a string containing the string representation of the two integers separated by a comma. Otherwise it produces a string representation of the head, followed by a comma, followed by a string representation of the tail. The tail must contain at least two elements, otherwise the pattern match would not reach that case.

3.2.1 Tail Recursion

However his function is not tail recursive! Lets rewrite it so it is:

```
fun inner_strlist (x: list[int], result:string) =>
  match x with
  // Empty list
  | #Empty => ""

  // One element list
  | Cons (head, #Empty) =>
    str head

  // Two element list
  | Cons (head, Cons (second, #Empty)) =>
    str head ", " + str second

  // More than two elements
  | Cons (head, tail) =>
    inner_strlist (tail, str head ", ")
  endmatch
;
```

Here is another, simpler tail recursive function:

```
fun addup (x: list[int]) =
{
  fun aux (rest: list, acc: int) =>
    match rest with
    | #Empty => rest
    | Cons (head, tail) => aux (tail, acc + head)
    endmatch
  ;
  return aux x 0;
}
```

The non-recursive function `addup` contains a nested tail recursive function `aux`.

3.2.2 Higher Order Functions

Ok so, what if I wanted to add up the squares of the numbers in the list? Is there a way to avoid rewriting the whole thing? The answer is yes:

```
fun fold
(
  binop: int * int -> int,
  init: int,
  lst: list[int]
) =
{
  fun aux (rest: list[int], acc: int) =>
    match rest with
    | #Empty => rest
    | Cons (head, tail) => aux (tail, binop (acc, head))
    endmatch
  ;
  return aux x init;
}
```

Fold works with any binary operator on `int`. For example:

```
fun addition (acc:int,y:int) => x + y;
assert addup x == fold (addition, 0, x);
```

There is another way to write the folding expression:

```
assert addup x == fold (fun (acc:int, y:elt)=>x+y), 0, x);
```

using an unnamed, or anonymous function, sometimes called a lambda.

So what happens if we want to make the sum of the square roots of a list of `int`?

It should work, right, if we change the signature of the binary operator to

```
double * int -> double
```

and pass an appropriate function. The first component of the argument is intended to hold the accumulated sum.

What if the list were of `string` and we wanted to get the complete length? Then we'd use

```
int * string -> int
```

3.2.3 Polymorphism

Can we generalise, so we only have one function?

Sure. That's what parametric polymorphism is for. Using R for the result type and V for the list element type:

```
fun fold[V,R]
(
  binop: R * V -> R,
  init: R,
  lst: list[V]
) =
{
  fun aux (rest: list[V], acc: R) =>
    match rest with
    | #Empty => rest
    | Cons (head, tail) => aux (tail, binop (acc,head))
  endmatch
;
  return aux (lst, init);
}
```

Now we can write:

```
println$
  fold [string,int]
  (
    (fun (acc:int, s:string)=>acc + len s),
    0,
    (list ("Hello", "World"))
  )
;
```

This is all very nice! Even better we can write:

```
println$
  fold
  (
    (fun (acc:int, s:string)=>acc + len s),
    0,
    (list ("Hello", "World"))
  )
;
```

because the instances of the type variables required can be deduced from the arguments.

However our implementation has a slight ugliness: we first define a function, then return an application of it, which consists of two statements, and requires the expanded form of function definition.

3.2.4 Let/in construction

Can we do it as a one liner, using only the simplified form? The answer is yes, using a `let-in` expression:

```
fun fold[V,R]
(
  binop: R * V -> R,
  init:R,
  lst:list[V]
) =>
  let fun aux (rest: lst, acc: R) =>
    match rest with
    | #Empty => rest
    | Cons (head, tail) => aux (tail, binop (acc,head))
  endmatch
  in
  aux lst init
;
```

The fold above utilises what is called a *type schema*. This is a limited kind of polymorphism also called first order polymorphism, or simply *templates*. Type schemes are limited to having type variables universally quantified on the outside left, and can be instantiated by replacing the type variable with an actual type. At run time, there are no type variables about. You cannot have a polymorphic function at run time. So

```
// NO polymorphic function closures in Felix!
type-error var x = fold[int, string];
```

There is another way to write this, using a method called currying, in a form called curry form. This is the more conventional form in some languages.

In functional programming languages, functions are first class, meaning you can pass them into another function, and return them from a function. Consider this:

```

fun fold[V,R] (binop: R * V -> R) =
{
  fun A(init:R)
  {
    fun B(lst:list[V])
    {
      let fun aux (rest: list[V], acc: int) =>
        match rest with
        | #Empty => rest
        | Cons (head, tail) => aux (tail, binop (acc,head))
      endmatch
      in
        return aux lst init;
      }
    return B;
  }
  return A;
}

```

What is the type of this function? Well,

```
B: list[V] -> R
```

and since A returns B, its type is:

```
A: R -> (list[V] -> R)
```

and since fold returns A its type is

```
fold: list[V] -> (R -> (list[V] -> R))
```

How would we use this? Well:

```

println$
(
  (
    (
      fold
        (fun (acc:int, s:string)=>acc + len s))
    )
    0
  )
  (list ("Hello", "World"))
)
;

```

Note that the `->` function operator is right associative we don't need the parens so we can write this as:

```
fold: list[V] -> R -> list[V] -> R
```

getting rid of parens, and, since application (the whitespace operator!) is left associative, we can write:

```
println$
  fold
    (fun (acc: int, s: string) => acc + len s))
    0
    (list ("Hello", "World"))
;
```

again getting rid of parens. The form:

```
fold fn init alist
```

is the curried form of the call we wrote in tuple form like:

```
fold (fn, init, alist)
```

Curried form has an advantage:

```
fun string_lengths (x:list[string]) =>
  fold
    (fun (acc:int, s:string)=>acc + len s))
    0
;

var l1 = list ("Hello", "World");
var l2 = list ("Felix", "Rocks");
println$ string_lengths l1, string_lengths l2;
```

In the function, we simply apply the fold to the first three arguments, which returns a function of one argument, a list of strings.

Curried form is so useful, there is syntactic sugar for writing functions in this form:

```
fun fold[R, V]
  (binop: R * V -> R)
  (init: R)
  (lst: list[V])
=>
  let fun aux (rest: list[V], acc: R) =>
    match rest with
    | #Empty => rest
    | Cons (head, tail) => aux (tail, binop (acc,head))
  endmatch
  in
  aux x init
;
```

3.3 Generalised Folds: Catamorphisms

Unsurprisingly, folds can be mechanically written for many data types. A more precise formulation in category theory can be given, and is explained by the notion of a [catamorphism](#). We will come back to this later.

Let us develop a tree fold. Our tree is binary and carries data at each node: both leaf and branch nodes:

```
union tree[T] =
  | Leaf of T
  | Branch of T * tree[T] * tree[T]
;

fun fold[R,T]
  (binop: R * V -> R)
  (init: R)
  (tr: tree[T])
=>
  match tr with
  | Leaf v => binop (acc,v)
  | Branch (v, l, r) =>
    let left = fold binop v l in
    let right = fold binop left r in
    binop (acc, right)
;
```

Note this formulation isn't tail recursive: it's not possible to visit tree nodes without remembering where you're up to.

3.4 Maps

Actually maps are easier than folds! A map function simply copies the structure of its argument, applying a unary operator to each element. Here is a list map:

```
fun map[R,V] (f: V -> R) (lst: list[V]) : list[R] =>
  match lst with
  | #Empty => Empty[R]
  | Cons (head, tail) => Cons (f head, map f tail)
  endmatch
;
```

You will note this implementation is not tail recursive! We will leave it as an exercise for the reader to develop a tail recursive formulation.

Now we will do a tree map:


```

fun map[R,V] (f: V -> R) (tr: tree[V]) : tree[R] =>
  match lst with
  | Leaf v => Leaf (f v)
  | Branch (v,l,r) => Branch (f v,map f l, map f r)
endmatch
;

```

As you can guess, writing maps is rather boring. This is because a map produces an exact copy of the original data structure with only the data type changed. An operation which *preserves structure* has a technical name: it is called a **functor**.

We will provide a way to encode maps in a type class now, but before we can do this we need to observe a technical detail!

Although our list and tree are considered in the abstract as data functors, in fact they are type schema, or indexed types. They're not actually functors. So to make them functors we will have to do this:

```

typedef list_f = fun (T:TYPE): TYPE => list[T];
typedef tree_f = fun (T:TYPE): TYPE => tree[T];

```

This is our first example of *higher kinded programming* which is sometimes loosely called meta-programming. The functions we have shown takes a type and produces another type. For example:

```

typedef intlist = list_f int;

```

Note that `list_f` is not yet a true functor. It is a map from types to types which is the object part of a functor definition. The collection of types forms a category which is named `TYPE` in Felix.

Now we can rewrite our maps like this:

```

class Functor[F:TYPE->TYPE]
{
  virtual fun fmap[V,R]: (V->R) -> (F V -> F R);
}

instance Functor[list_f]
{
  fun fmap[V,R] (f:V->T) (lst: list_f V) : list_f R=>
    match lst with
    | #Empty => Empty[R]
    | Cons (head, tail) => Cons (f head, fmap f tail)
    endmatch
  ;
}

instance Functor[tree_f]
{
  fun fmap[V,R] (f: V -> R) (tr: tree[V]) : tree[R] =>
    match tr with
    | Leaf v => Leaf (f v)
    | Branch (v,l,r) => Branch (f v,fmap f l, fmap f r)
    endmatch
  ;
}

```

The pair consisting of the type mapping `list_f` and an associate function mapping `Functor[list_f]::fmap` are a functor, because the two constraints are satisfied. First, it's obvious that for any type T , an identity map:

```
fun id(x:T) => x;
```

is taken to the function

```
Functor[list_f]::fmap id
```

This just makes a copy of the list. Secondly, given any two function

```

fun f[A,B]: A -> B;
fun g[B,C]: B -> C;

```

then:

```
(Functor[list_f]::fmap g) \circ (Functor[list_f]::fmap f)
```

has the same semantics as:

```
Functor[list_f]::fmap (g \circ f)
```

In other words, mapping a list using f and then mapping it again using g produces the same final result as mapping it by the composite function $g \circ f$ in one

go. This requirement is more than just a rule, however. Its a crucial optimisation called [deforestation](#) or *fusion*. Although not always the case, applying the composite function once to each element to build a new list is probably faster than building a temporary list, building another from it, and the deleting the temporary list.

It semantic rules for functors are important enough to encode them in the type class:

```
class Functor[F:TYPE->TYPE]
{
  virtual fun fmap[V,R]: (V->R) -> (F V -> F R);

  axiom identity_preserving[A,B] (f:A->B) (id:A->A) (x:A):
    id x == x \implies fmap id (f x) = fmap f x
  ;

  reduce structure_preserving[A,B,C] (f:A->B, g:B->C):
    (fmap g) \circ (fmap f) =>
    fmap (g \circ f)
  ;
}
```

The identity rules is a bit tricky, it says that for every function $f : A \rightarrow B$ and function $id : A \rightarrow A$ for which for all $x \in A$ we have $fx = x$, then the `fmap` of the `id` function applied to the value `f` maps `x` to is the same as the value that the `fmap` of function `f` gives. We need this long winded exposition because we cannot directly assert that two functions are equal.

The current version of Felix can check the axioms against test data, however it cannot mechanically implement the reduction rule to effect an optimisation.

Part III

Procedural Programming

Chapter 4

Procedural Basics

Chapter 5

Dynamic Objects

Chapter 6

Reactive Programming

Part IV

Active Programming

Chapter 7

Fibres

Chapter 8

Synchronous Channels

Chapter 9

Pipelines

Chapter 10

Coroutines

Chapter 11

Continuations

Chapter 12

Symmetric Lambda Calculus

Part V

Concurrency

Chapter 13

Asynchronous Events

Chapter 14

Pre-emption

14.1 Pthreads

14.1.1 Spawning Pre-emptive Threads

14.1.2 Atomics

14.1.3 Locks

14.1.4 Barriers

14.1.5 Concurrently Statement

14.2 Pchannels

Chapter 15

Parallelism

15.1 Thread Pool

15.2 Parallel For Loop

Part VI

Categorical Programming

Chapter 16

Category Theory

Chapter 17

Symmetric Categorical Language

Chapter 18

Meta-programming

Chapter 19

Polyadicity: The Holy Grail